Question: I use this code to find the `ConvexHull` for some points:

```
pts2 = {{2, 1}, {5, 0}, {7, 0}, {9, 5}, {6, 8}, {0, 6}, {0, 4}};
Show[ConvexHullMesh[pts2, GridLines -> Automatic], Axes -> True]

```

the result is this pic:

[![enter image description here](https://i.stack.imgur.com/Hmx2x.png)](https://i.stack.imgur.com/Hmx2x.png)

I need the result like this:

[![enter image description here](https://i.stack.imgur.com/N2VSw.png)](https://i.stack.imgur.com/N2VSw.png)

Answer: 
**Update**

*(included more style specifications as requested in the comments)*

You can use [`BaseStyle`](http://reference.wolfram.com/language/ref/BaseStyle) to control the appearance of the convex hull, and you can extract the coordinates of a mesh by using `mesh["Coordinates"]`. Finally, [`Arrowheads`](http://reference.wolfram.com/language/ref/Arrowheads) can be used to style the axes:

```
pts2={{2,1},{5,0},{7,0},{9,5},{6,8},{0,6},{0,4}};

mesh = ConvexHullMesh[
    pts2,
    BaseStyle->{FrontFaceColor->None,EdgeColor->Black,FrontFaceOpacity->0}
];
Show[
    mesh,
    Graphics[{PointSize[Large],Point[mesh["Coordinates"]]}],
    GridLines->Automatic,
    Axes->True,
    AxesStyle->Arrowheads[{0.0,0.05}]
]

```

[![enter image description here](https://i.stack.imgur.com/Fqo3Y.png)](https://i.stack.imgur.com/Fqo3Y.png)
Here is an approximate facsimile:

```
ch = MeshPrimitives[ConvexHullMesh[pts2], 2];
bb = BoundingRegion[pts2];
cent = Mean[pts2];

Graphics[{{EdgeForm[Directive[Dashed, AbsoluteThickness[1]]], FaceForm[], bb},
          {EdgeForm[Black], FaceForm[], ch},
          {AbsolutePointSize[6], Point[pts2]},
          {Text[Style["Γ", FontFamily -> "Times", FontSize -> 14], cent],
           Text[TraditionalForm[Subscript["O", Row[{0, 0}]]], {0, 0}, {1.2, 1.2}],
           Text[TraditionalForm[Subscript["O", Row[{∞, ∞}]]], Last[bb], -{1.2, 1.2}]},
          {MapIndexed[Text[Subscript["γ", #2[[1]]], cent + 0.9 (Mean[#] - cent)] &, 
                      Partition[ch[[1, 1]], 2, 1, 1]]}},
         Axes -> True, AxesLabel -> {Style["μ", Medium], Style["ν", Medium]}, 
         AxesStyle -> Arrowheads[{0., 0.05}],
         GridLines -> Automatic, GridLinesStyle -> Directive[Dotted, Opacity[1/4]],
         Method -> {"AxesInFront" -> False, "GridLinesInFront" -> False},
         PlotRangePadding -> Scaled[0.1],
         Ticks -> {{{bb[[2, 1]], Subscript["O", Row[{∞, 0}]]}},
                   {{bb[[2, 2]], Subscript["O", Row[{0, ∞}]]}}}]

```

![figure](https://i.stack.imgur.com/g2FcD.png)